

```
/**
**
** * MACRO: MACEDIT.WCM
** * PURPOSE: Edit a small macro in a PRWin text box
**
** */
```

```
Application (A1; "WpprWin"; Default; "US")
InhibitInput(On!)
```

```
/**
**
** * VARIABLES - Declaring and Assigning Variables
**
** */
vCompile:=1 //Compile the Macro on save?
vMacroPath:=EnvPaths(PathType:Macros!) //Default macros path
vMacroName:="" //default macro name
vDialogResult=0 //get results from macro dialogs
vControl=0 //control save loop
```

```
/**
**
** * DIALOG DEFINITION
** * Edit Macro Dialog - Get Macro To Edit
**
** */
DIALOGDEFINE(100; 50; 50; 240; 68;Percent!; "Edit Macro")
    DIALOGADDTEXT(100; 101; 10; 10; 75; 15; Left!; "&Filename")
    DIALOGADDFILENAMEBOX(100; 102; 10; 20; 150; 15; FilesAndDirs!;
vMacroName; vMacroPath; "*.wcm")
    DIALOGADDPUSHBUTTON(100; 103; 175; 10; 50; 13; DefaultBttn!; "OK")
    DIALOGADDPUSHBUTTON(100; 104; 175; 28; 50; 13; NonDefaultBttn!; "Cancel")
```

```
/**
**
** * DIALOG DEFINITION
** * Save Macro Dialog - Save the Macro being edited
**
** */
DIALOGDEFINE(300; 98; 2; 105; 66; Percent! | Modeless!; "Macro Edit")
    DIALOGADDTEXT(300; 301; 3; 3; 90; 15; Center!; "Click in the Text to Edit,")
    DIALOGADDTEXT(300; 302; 3; 11; 90; 15; Center!; "click Continue to save the")
    DIALOGADDTEXT(300; 303; 3; 19; 90; 15; Center!; "macro and close the editor.")
    DIALOGADDPUSHBUTTON(300; 304; 22; 31; 60; 13; DefaultBttn!; "Continue")
```

```
/**
**
** * DIALOG DEFINITION
** * Save Macro Dialog - Save the Macro being edited
**
** */
```

```

DIALOGDEFINE(200;50;50;228;65;Percent!;"Save Macro")
    DIALOGADDTEXT(200; 201; 7; 7; 150; 15; Left!; "&Filename")
    DIALOGADDFILENAMEBOX(200 ;202; 7; 17; 150; 15; FilesAndDirs!; vMacroName;
vMacroPath; "*.wcm")
    DIALOGADDPUSHBUTTON(200; 203; 165; 7; 50; 13; DefaultBtn!; "OK")
    DIALOGADDPUSHBUTTON(200; 204; 165; 25; 50; 13; NonDefaultBtn!; "Cancel")

```

```

//start macro
//display dialog to get filename
LABEL(Start)
DIALOGSHOW(100;"WpprWin")
vDialogResult=MacroDialogResult
DIALOGDISMISS(100;103)
IF(vDialogResult=104)
    QUIT
ENDIF

```

```

//check validity of macro filename and if not valid get a new name
CALL(CheckFileName@)
WHILE(NOT DoesFileExist(vMacroName))
    MESSAGEBOX(vDialogResult; "Macro Edit"; "Macro File Not Found";
        IconExclamation!)
    DIALOGSHOW(100;"WpprWin")
    vDialogResult=MacroDialogResult
    DIALOGDISMISS(100;103)
    IF(vDialogResult=104)
        QUIT
    ENDIF
    CALL(CheckFileName)
ENDWHILE

```

```

IF(FileSize(vMacroName)>64000)
    MESSAGEBOX(vDialogResult; "Macro Edit"; "The macro "+vMacroName+" is larger
than the text editor can handle, use WordPerfect to edit large macros. Part of the macro may be
lost if you continue. Continue?"; IconExclamation! | YesNoCancel!)
    SWITCH(vDialogResult)
        CASEOF 2: //cancel
            QUIT
        CASEOF 7: //no
            GO(Start@)
        CASEOF 6: //yes Continue
    ENDSWITCH
ENDIF

```

```

//start editing the macro

```

```

NewDrawingCreate()
TextBoxCreate ()
AddTextBox (617w; 9575w; 12600w; 8975w)
//set font smaller so more macro fits in the box
SetFontFace ("Arial Regular"; 15; ; ; ; ; TT!)
SetFontPattern (Solid!; 0; 0; 0; 255; 255; 255; 0; 0; 0; 0; 0; 0; 0; 0; FramedFilled!)
SetFontGradient (None!; 64; 0; 0; 0; 2; 0; 0; 0; 255; 255; 255; 255; 255; 255; 0; 0)
RealizeFontChanges (Selected!)
//retrieve the macro
FileRetrieve (vMacroName; YES!)
InhibitInput(Off!)

//show a modeless dialog waiting for the user to finish editing
DIALOGSHOW(300;"WpprWin")
DIALOGDISMISS(300;304)

//after editing save the macro
WHILE(vControl=0)
    //get the name to save the macro as
    DIALOGSHOW(200;"WpprWin")
    vDialogResult:=MacroDialogResult
    DIALOGDISMISS(200;203)

    //check output of the dialog
    SWITCH(vDialogResult)
        //if cancel was selected see if they really want to lose changes
        CASEOF 204:
            MESSAGEBOX(vDialogResult;"MacroEdit";"Cancel Changes to
Macro?";YesNo! | IconQuestion!)
            SWITCH(vDialogResult)
                CASEOF 6:
                    vControl=1 //yes
            ENDSWITCH
        //if OK was selected save the file
        CASEOF 203:
            FileSaveAs (vMacroName; WPPresentation_20!; TRUE; FALSE)
            vControl=1
    ENDSWITCH
ENDWHILE

//close the text box and the drawing screen
TextEditExit ()
FileClose (NO!)
QUIT

```

```
//check the filename entered for correct path and extension
LABEL(CheckFileName)
  IF(STRLEN(vMacroName)<12)
    IF(SUBSTR(vMacroName;2;1) <> ":")
      ASSIGN(vMacroName;vMacroPath+vMacroName)
    ENDIF
  ENDIF
  IF(SUBSTR(vMacroName;STRLEN(vMacroName)-3;4) <> ".wcm")
    ASSIGN(vMacroName;vMacroName+".wcm")
  ENDIF
RETURN
```